
Abstract. Sorting algorithms are one of the key pedagogical foundations of com-
puter science, and their properties have been studied heavily. Perhaps less well
known, however, is the fact that many of the basic sorting algorithms exist as
a pair, and that these pairs arise naturally out of the duality between folds and
unfolds. In this paper, we make this duality explicit, by showing how to define
common sorting algorithms as folds of unfolds, or, dually, as unfolds of folds.
This duality is preserved even when considering optimised sorting algorithms
that require more exotic variations of folds and unfolds, and intermediary data
structures. While all this material arises naturally from a categorical modelling
of these recursion schemes, we endeavour to keep this presentation accessible to
those not versed in abstract nonsense.

1 Introduction

Sorting, described in great detail by Knuth (1998), is one of the most important and
fundamental concepts in computer science. In one form or another, sorting appears in
nearly every domain of computer science. As such, there are many different implemen-
tations of sorting algorithms, with varying performance and complexity.

One of the simplest sorting algorithms is insertion sort, which revolves around the
idea of inserting a single element in an already sorted list. To sort a list of elements
using this strategy, we take the next element in the list that is to be considered, insert
it into an accumulated sorted list that is initially empty, and proceed recursively until
all elements have been inserted. In Haskell (Peyton Jones et al. 2003), insertion sort is
concisely expressed using the foldr operation on lists, defining it as the application of
the insert operation to each element of the input list, producing a result starting with the
empty list:

insertSort :: [Integer ]→ [Integer]
insertSort = foldr insert [ ]

The insert function takes one element and inserts it in an already sorted list. It does this
using span to break the sorted list into two segments according to the pivot element we
want to insert, which is then introduced in between the two parts:

insert :: Integer → [Integer ]→ [Integer]
insert y ys = xs++[y ]++ zs

where (xs,zs) = span (� y) ys

The use of span relies on the fact that the list ys is already sorted.
Another basic sorting algorithm, selection sort, can be easily expressed in terms

of an unfold. Unfolds are a recursion scheme dual to folds, and are used to produce
data, instead of consuming data. Unfolds are often given less attention than folds
(Gibbons and Jones 1998), but that is not the case in this paper: we assure the reader
that we will maintain proportional representation, and show an unfold for every fold. In
Haskell, the unfold operation on lists is defined as unfoldr:

unfoldr :: (b → Maybe (a,b))→ (b → [a ])

The first argument to unfoldr defines how to produce lists from a seed: the Nothing
case corresponds to the empty list, whereas Just (a,b) corresponds to a list with
element a and new seed b. Using this function, and a starting seed, unfoldr produces a
complete list.

Selection sort works on a list by recursively picking the smallest element of the input
list and adding this element to the result list. It can be defined as the unfold of a select
operation:

selectSort :: [Integer ]→ [Integer ]
selectSort = unfoldr select

This select operation picks the smallest element from the input list using minimum,
removes it using delete, and continues recursively by returning the smallest element
together with the remaining list.

select :: [Integer]→ Maybe (Integer, [Integer ])
select [ ] = Nothing
select xs = Just (x,xs′)

where x = minimum xs
xs′ = delete x xs

For practical reasons, the types of foldr and unfoldr in Haskell are not clearly dual.
This contributes to obscuring the inherent duality in our insert and select functions.
The purpose of this paper is to explicitly highlight the duality in sorting algorithms, so
that we can provide a unified definition for both insertSort and selectSort. We do this by
exploring a type-directed approach to algorithm development, where the types dictate
most of the behaviour of functions.

The remainder of this paper is structured as follows. We first introduce a framework
of functors, folds, and unfolds in Section 2, which we use in Section 3 to implement
two exchange sorts in one go. To define more efficient insertion and selection sorts, we
start by introducing more exotic variants of folds and unfolds, called paramorphisms
and apomorphisms, in Section 4. We use these morphisms in Section 5, revisiting the
two sorting algorithms shown in this introduction. In Section 6 we turn our attention to
mergesort, in order to show how these recursion schemes can be applied to create more
efficient sorting algorithms. We conclude our discussion in Section 7.

This paper is built on our earlier work on this same subject (Hinze et al. 2012), but we
have rewritten the exposition entirely, simplifying many aspects and removing all the

category theory jargon. The theoretically-inclined reader is referred to the earlier work
for a deeper understanding of bialgebras and distributive laws in sorting, but this is not
required for the comprehension of this paper; the entire development arises naturally
out of a type-directed approach to programming, without need for appealing to category
theory for the justification of design choices.

2 Functors, Folds, and Unfolds

In this paper we focus on the duality of folds and unfolds, and how these recursion
schemes can be used in sorting. The standard functions foldr and unfoldr are particu-
larly useful since they allow us to express a whole class of recursive functions. Their
utility draws from the fact that folds and unfolds allow us to abstract away from using
functions with direct recursion. In the case of folds, the exact site of the recursive step is
handled by the foldr function, and the non-recursive component is described by its pa-
rameters, which constitute a so-called algebra. Dually, unfolds are considered in terms
of a corecursive step and a non-recursive coalgebra.

This pattern is mirrored at the level of data and is not unique to lists, where recur-
sive datatypes are described as two-level types (Sheard and Pasalic 2004): one level
describes the fact that the data is recursive, and the other is non-recursive and describes
the shape of the data. Using this representation we can decompose the list datatype into
two parts. First, consider the non-recursive component:

data List list = Nil | Cons Integer list

We call a datatype such as List the base functor of the recursive type. For simplicity
we consider lists of elements of type Integer; our development generalises readily to
polymorphic lists with an Ord constraint on the element type.

Note that the type of List is intrinsically not recursive, but instead uses a parameter
where one might expect the recursive site. We can retrieve the usual lists from our non-
recursive List datatype using the fixed-point combinator Fix, which builds recursion
into datatypes:

newtype Fix f = In {out :: f (Fix f )}
Combining these two parts into a two-level type yields Fix List, which is isomorphic to
the predefined type of integer lists [Integer].

The recursive component of this data structure is marked by the Functor instance of
the base functor, and is key to providing a generalised definition of a fold:

instance Functor List where
fmap f Nil = Nil
fmap f (Cons k x) = Cons k (f x)

Note that this is not the same functoriality as the one typically used to express a mapping
over the elements in a list.

The advantage of representing lists by their base functor becomes evident when
we define the fold and unfold operations. Datatypes defined by abstracting over the
recursive positions, like List, enjoy a single fold operator:

fold :: (Functor f )⇒ (f a → a)→ (Fix f → a)
fold a = a · fmap (fold a) ·out

The fold function takes an argument a, called the algebra, that is able to crush one level
of the data structure. The definition works by first exposing the top level of the structure
by using out, which is then crushed at its recursive sites using fmap, and finally crushed
at the top level using a.

There is also a single, generic definition of unfold, which is now clearly dual to fold:

unfold :: (Functor f )⇒ (a → f a)→ (a → Fix f )
unfold c = In · fmap (unfold c) ·c

The first argument to unfold, the coalgebra c, defines how to expand a seed into some
functorial type f with seeds at the leaves. The coalgebra is applied by unfold recursively
until a complete structure is built. Again the recursive site is marked by the Functor
instance of the structure in question.

Unlike Haskell lists, which have foldr and unfoldr operations specialised to their
type, our fold and unfold operations work on any datatype with a Functor instance,
and we will soon make use of this generality. We have only sketched the details of
base functors and their recursive morphisms; a more detailed presentation, including
relevant category theory background, can be found in Meijer et al. (1991). Another cat-
egorical treatment that generalises folds and unfolds to operate on an even wider class
of recursive types can be found in Hinze (2011).

3 Sorting by Swapping

In this section we will look at our first sorting algorithms expressed in terms of the
generalised folds and unfolds introduced in the previous section, and show how duality
naturally arises in this setting. To ease the understanding of the algebras and coalgebras
that we will see, which generally perform “one step” of sorting, we introduce a datatype
of sorted lists, together with its Functor instance:

data List list = Nil | Cons Integer list

instance Functor List where
fmap f Nil = Nil
fmap f (Cons k list) = Cons k (f list)

Note that List is entirely isomorphic to List. The only difference lies in the names used:
the fact that a list is sorted is indicated by the underlining on its type and constructor
names. The compiler will not be able to enforce the condition that List always represents
sorted lists for us, but we keep this invariant throughout our development.

A sorting algorithm, in general, takes arbitrary lists to sorted lists:

sort :: Fix List → Fix List

By looking at its type, we can interpret sort as either a fold, that consumes a value of
type Fix List, or as an unfold that produces a value of type Fix List. If sort is a fold,



A Duality of Sorts 155

its algebra will have type List (Fix List)→ Fix List. This algebra can then be defined as
an unfold which produces a value of type Fix List. These observations are summarised
in the following type signatures:

fold (unfold c) :: Fix List → Fix List

unfold c :: List (Fix List)→ Fix List

c :: List (Fix List)→ List (List (Fix List))

Using this approach brings an additional benefit: the analysis of the complexity of our
algorithms can be framed in terms of the cost of the fold and unfold functions. The
running time of a fully evaluated result of a fold is proportional to the depth of its input
structure multiplied by the cost of one step of the algebra. Dually, the running time
of an unfold is proportional to the depth of its output structure multiplied by the cost
of one step of the coalgebra. This property will become useful when evaluating the
performance of our algorithms.

Let us then write a sorting function as a fold of an unfold:

naiveInsertSort :: Fix List → Fix List
naiveInsertSort = fold (unfold naiveInsert)

naiveInsert :: List (Fix List)→ List (List (Fix List))
naiveInsert Nil = Nil
naiveInsert (Cons a (In Nil)) = Cons a Nil
naiveInsert (Cons a (In (Cons b x)))

| a � b = Cons a (Cons b x)
| otherwise = Cons b (Cons a x)

Most of the behaviour of naiveInsert follows naturally from its type. The empty and
single element unsorted lists are trivially converted into sorted variants. For an unsorted
list with at least two elements, we compare the elements, reordering if necessary. What
we obtain is a form of “naive” insertion sort, since it does not make use of the fact that
the list where an element is being inserted in is already sorted. Instead, the traversal is
continued, even though there is no more work to be done. Indeed, the analysis of the
time complexity of this algorithm is simple: the input size of the fold is linear, and the
output size of the inner unfold is also linear, so we should expect quadratic behaviour.
We will see how to make use of the fact that the inner list is already sorted in Section 5.

Recall now that we can also see a sorting function as an unfold of a fold. In that case,
the type of the inner algebra can be derived as in the following type signatures:

unfold (fold a) :: Fix List → Fix List

fold a :: Fix List → List (Fix List)

a :: List (List (Fix List))→ List (Fix List)

The sorting algorithm that we obtain as an unfold of a fold is a version of bubble sort:

bubbleSort :: Fix List → Fix List
bubbleSort = unfold (fold bubble)

bubble :: List (List (Fix List))→ List (Fix List)
bubble Nil = Nil
bubble (Cons a Nil) = Cons a (In Nil)
bubble (Cons a (Cons b x))

| a � b = Cons a (In (Cons b x))
| otherwise = Cons b (In (Cons a x))

This algorithm proceeds by continually comparing adjacent elements, swapping them
if they are in the wrong order, which is the principal idea behind a bubble sort. The
similarity between bubble and naiveInsert is striking; they differ only in the placement
of the fixed-point constructor In. This becomes clear if we look at their types, after
expanding one definition of Fix in each of them:

naiveInsert :: List (List (Fix List))→ List (List (Fix List))
bubble :: List (List (Fix List))→ List (List (Fix List))

The only difference is in the inner type of lists at the third level of depth. However, this
third level is in some sense redundant, since these algorithms only inspect elements in
the first two levels. It is this observation that allows naiveInsert and bubble to be safely
generalised to a step function of the following type:

swap :: List (List x)→ List (List x)

Such a step function is sometimes called a distributive law, since it captures an abstract
notion of distributivity. The definition of this new function, which we call swap since it
simply swaps adjacent elements based on their order, is entirely similar to the definitions
of both naiveInsert and bubble:

swap Nil = Nil
swap (Cons a Nil) = Cons a Nil
swap (Cons a (Cons b x))

| a � b = Cons a (Cons b x)
| otherwise = Cons b (Cons a x)

The swap function can be understood as a distributive law between the types List and
List, and is a generalisation that captures the essence of both naiveInsert and bubble.
We can use swap to define both naiveInsertSort′ and bubbleSort′:

naiveInsertSort′,bubbleSort′ :: Fix List → Fix List

naiveInsertSort′ = fold (unfold (swap · fmap out))
bubbleSort′ = unfold (fold (fmap In · swap))

The use of fmap out in naiveInsertSort′, and, dually, fmap In in bubbleSort′, reflects our
expansion of the Fix datatype in the type of the (co)algebra. What we have obtained is a
single definition for two conceptually distinct sorting algorithms, in terms of a function
that expresses how to perform one step of the computation.

At this point it is worth reinforcing our intuition for how these algorithms work.
The duality of these sorting algorithms can be seen visually when we assume a call-by-
value evaluation order of the definitions. The diagrams below emphasise that the actual

comparisons performed by swap are the same, and that the algorithms only differ in the
order in which these comparisons are performed:

naive insertion sort
input

2 5 4 1 3
2 5 4 1 3 1↔3
2 5 4 1 3 4 ↔̇1 4 ↔̇3
2 5 1 3 4 5 ↔̇1 5 ↔̇3 5 ↔̇4
2 1 3 4 5 2 ↔̇1 2↔3 3↔4 4↔5
1 2 3 4 5

output
bubble sort

input
2 5 4 1 3
1 2 5 4 3 2 ↔̇1 5 ↔̇1 4 ↔̇1 1↔3
1 2 3 5 4 2↔3 5 ↔̇3 4 ↔̇3
1 2 3 4 5 3↔4 5 ↔̇4
1 2 3 4 5 4↔5
1 2 3 4 5

output

For both of these algorithms, it is the outer recursion scheme that drives the computa-
tion. In the case of naiveInsertSort′ this is a fold, and the progression is depicted by the
vertical line that separates sorted from unsorted data working its way from the end of the
unsorted list until only a sorted list remains. Since bubbleSort′ is expressed as an unfold,
the corresponding diagram is pleasingly dual. The computation starts from the beginning
of a sorted list, beginning with an empty list and gradually bubbling values to the front.
Notice how for naiveInsertSort′, the input remains stable, whereas the output does not,
whereas for bubbleSort′, it is the output that remains stable, whereas the input does not.

On the right side of these diagrams we have presented the comparisons that take place
in the swap function, which are effectively determined by the inner recursion. The ar-
rows correspond to comparisons that are made between two elements, and dotted arrows
indicate comparisons that result in a swap (this is the case when the element on the left is
greater than that on the right). The arrows that correspond to each line of naiveInsertSort′
show the comparisons that are needed to insert the element immediately to the left of the
vertical line, and should be read from left to right. On the other hand, the swaps that
correspond to each line of bubbleSort′ show the comparisons that are needed to select
the value to the left of the vertical line, and should be read from right to left.

4 Paramorphisms and Apomorphisms

In the previous section we defined two sorting algorithms with performance Θ(n2),
where both work in quadratic time regardless of the input. In particular, the unfolding
step in the insertion sort will continue to traverse a sorted structure long after a new
element has been inserted into its appropriate place. This is unfortunate, as it fails to

make use of the fact that the output that has already been constructed is sorted. If this
information was taken into consideration, the inner traversals could be interrupted for
better performance. However, using folds and unfolds as recursion schemes prohibits
us from meddling with the recursion, and no such interruption is possible. In order to
gain explicit control of when an unfold should stop traversing a structure, we turn to
a slightly more exotic version of unfold, namely the apomorphism (Vene and Uustalu
1998), which gives us the required ability to abort recursion. Dually, we will also make
use of paramorphisms (Meertens 1992), which give us the power to use a part of the
input data in an algebra. Paramorphisms and apomorphisms can be understood as the
counterparts to folds and unfolds, and enjoy aspects of duality.

Before diving into the details of these recursion schemes, however, it is informative
to first consider the duality that exists between so-called product and sum types. The
product of types is simply a synonym for a pair of values, and the sum of types is a
synonym for the Either type (a choice between two values):

type a× b = (a,b)
type a+ b = Either a b

In a sense, these operators encode a form of arithmetic on types. Assuming a is a type
with m inhabitants, and b a type with n inhabitants, the product type a× b is inhabited
by m× n values, as we can choose one element of b for each element of a. Similarly,
the sum type a+ b has m+ n inhabitants, as we have to pick either an element from a
or an element from b.

The duality between these types can be understood in terms of two operators: one
which constructs products, and another which deconstructs sums. The operator �, or
split, constructs a pair by applying two functions with a common source type:

(�) :: (x → a)→ (x → b)→ (x → a× b)
(f �g) x = (f x,g x)

The dual operator �, or join, deconstructs a sum by applying two functions with a
common target type:

(�) :: (a → x)→ (b → x)→ (a+ b → x)
(f �g) (Left a) = f a
(f �g) (Right b) = g b

Using these operators, we can extend the duality that folds and unfolds enjoy, and define
paramorphisms and apomorphisms.

A paramorphism is defined as a variation on fold which makes use of a product in
the source of its algebra. The product is used to “remember” the original Fix f structure:

para :: (Functor f )⇒ (f (Fix f × a)→ a)→ (Fix f → a)
para f = f · fmap (id�para f ) ·out

The first argument to para now has access to both the original Fix f structure and the
computed result for this same structure. This argument is not an algebra in the categor-
ical sense, but we shall name it so for simplicity, since it serves essentially the same
purpose. We will render the product constructor of two elements a and b as a b, to
remind us that the first element is the precomputed result of the second.

A typical example of a paramorphism on lists is the function that calculates all proper
suffixes of a list:

suffixes :: Fix List → [Fix List ]
suffixes = para suf

suf :: List (Fix List× [Fix List ])→ [Fix List ]
suf Nil = [ ]
suf (Cons _n (l ls)) = l : ls

Although it may seem like paramorphisms are more powerful than folds, this is not the
case. They simply make certain algorithms more convenient to express by providing
direct access to the original structure. This behaviour can also be expressed in a fold; in
fact, we can define para as a fold:

para′ :: (Functor f )⇒ (f (Fix f × a)→ a)→ (Fix f → a)
para′ f = snd · fold ((In · fmap fst)� f )

Dually, an apomorphism is a variation of an unfold which makes use of a sum in the
target of its first argument. As before, we abuse terminology, and name this argument
a coalgebra. This sum is used to encode the choice between stopping the apomorphism
with a concrete value of type Fix f , or going on with a new seed of type a:

apo :: (Functor f )⇒ (a → f (Fix f + a))→ (a → Fix f )
apo f = In · fmap (id�apo f ) · f

Here the coalgebra uses its source value to produce either a final result, or an interme-
diate step. If a final result is given then the recursion no longer continues; otherwise,
values are produced just as in an unfold. For mnemonic reasons, we will render the Left
constructor as Stop, as it encodes stopping the recursion, and Right as Go, as it encodes
continuing the traversal.

Note that the power to improve the running time of our sorting algorithms relies
on the use of apomorphisms. Paramorphisms are mostly a cosmetic improvement; the
resulting traversal still consumes the entire input linearly. Apomorphisms, on the other
hand, allow for early termination of the computation, so their running time is no longer
necessarily linear on the size of the resulting structure.

4.1 Folds of Apomorphisms, Unfolds of Paramorphisms

As before, we use a type-directed approach to guide the development of a sorting algo-
rithm, except this time we replace the inner recursions with apo and para, since we are
aiming for a more efficient algorithm. Deriving the appropriate algebra and coalgebra
yields the following:

fold (apo c) :: Fix List → Fix List

apo c :: List (Fix List)→ Fix List

c :: List (Fix List)→ List (Fix List+List (Fix List))

unfold (para a) :: Fix List → Fix List

para a :: Fix List → List (Fix List)

a :: List (Fix List×List (Fix List))→ List (Fix List)

The duality is somewhat hidden by some noise in the types, but can be easily re-
covered by introducing some synonyms for what are sometimes called pointed and
copointed types, and unrolling some fixpoints:

type f+ a = a+ f a
type f× a = a× f a

After unrolling one layer of the fixed point, we obtain the following types for our
(co)algebras:

c · fmap In :: List (List (Fix List))→ List (List+ (Fix List))
fmap out ·a :: List (List× (Fix List))→ List (List (Fix List))

From this it is clear that the (co)algebras are almost of the same form, except that the
coalgebra should be modified to consume a copointed type in its source, and the algebra
should be modified to produce a pointed type in its target.

This suggests that we can combine the (co)algebras into a single step function with
a more general type:

b :: List (List× x)→ List (List+ x)

For convenience, we shall abuse terminology, and occasionally call such step functions
distributive laws, since they serve almost the same purpose as the distributive laws in-
troduced in Section 3. With some gentle massaging, we can use such a step function in
the context of either an apomorphic coalgebra or a paramorphic algebra:

c = b · fmap (id�out) :: List (Fix List) → List (List+ (Fix List))
a = fmap (id� In) ·b :: List (List× (Fix List))→ List (Fix List)

Once again, the step function crucially depends on parametricity for unifying algebras
and coalgebras.

5 Insertion and Selection Sort

Now that we have apomorphisms, which allow us to stop recursion, we can write a
non-naive version of insertion sort that adequately stops traversing the result list once
the element has been inserted. Insertion sort is the fold of an apo:

insertSort :: Fix List → Fix List
insertSort = fold (apo insert)

The coalgebra insert is similar to naiveInsert, but with the essential difference that it
stops creating the list (with Stop) if no swapping is required. Otherwise, it continues
traversing (with Go):

insert :: List (Fix List)→ List (List+ (Fix List))
insert Nil = Nil
insert (Cons a (In Nil)) = Cons a (Stop (In Nil))
insert (Cons a (In (Cons b x′)))

| a � b = Cons a (Stop (In (Cons b x′)))
| otherwise = Cons b (Go (Cons a x′))

Because we are using apomorphisms, insertSort will run in linear time on a list that is
already sorted, as the inner traversal is immediately terminated each time it is started.
This behaviour is crucial for the best case behaviour of insertSort.

As before, we can find a dual algorithm to insertSort that is defined as an unfold of a
para. Instead of writing a specialised algebra, we will directly write the distributive law
that can be used both as argument to apo and para. Its type, as explained in Section 4,
is the following:

swop :: List (List× x)→ List (List+ x)

We nickname this function swop as it “swaps and stops”; its type indicates that it has
access to the sorted list as its argument, and that it can decide to abort recursion when
producing a result. Its definition is an unsurprising generalisation of insert:

swop Nil = Nil
swop (Cons a (x Nil)) = Cons a (Stop x)
swop (Cons a (x Cons b x′))

| a � b = Cons a (Stop x)
| otherwise = Cons b (Go (Cons a x′))

Having defined swop, we can use it to define an alternative version of insertSort, which
does not use insert:

insertSort′ :: Fix List → Fix List
insertSort′ = fold (apo (swop · fmap (id�out)))

However, being a distributive law, swop can also be used to construct the algebra of a
paramorphism .The sorting algorithm that we then obtain is selection sort:

selectSort :: Fix List → Fix List
selectSort = unfold (para (fmap (id� In) · swop))

Unlike bubble sort (the dual of “naive” insertion sort), selection sort uses the accumu-
lated result x in the a � b case, meaning the smallest element has been placed in the
correct location. We again get two, entirely dual sorting algorithms for the price of one
step function.

6 Mergesort

The work in the previous section brought us a slight boost in performance over the
naive version of insertion sort. However, its time complexity is still on average O(n2),

which is bound by the fact that we use folds and unfolds over lists: only the lower bound
was improved to Ω(n). To improve on the average bound we must move to a different
algorithm where an intermediate data structure with sublinear depth is built from the
input list, and then used to produce the output list. This two-phase approach was used in
our previous work to synthesise versions of quicksort and heapsort (Hinze et al. 2012).
In this section we show the development of mergesort, which improves the average case
complexity to Θ(n logn).

These two phases can be seen in a typical implementation of mergesort in Haskell,
where the recursive nature of the algorithm is expressed directly, rather than through an
intermediary datastructure:

mergeSort :: [Integer]→ [Integer]
mergeSort as = merge (mergeSort bs) (mergeSort cs)

where (bs,cs) = split as

split :: [Integer]→ ([Integer ], [Integer])
split [ ] = ([ ], [ ])
split [a ] = ([a ], [ ])
split (a : b : cs) = (a : as,b : bs)

where (as,bs) = split cs

merge :: [Integer]→ [Integer]→ [Integer ]
merge as [ ] = as
merge [ ] bs = bs
merge (a : as) (b : bs)

| a � b = a : merge as (b : bs)
| otherwise = b : merge (a : as) bs

In the first phase, split is called at each recursive step of mergeSort, and recursively
splits the input list in two by uninterleaving the elements. The merge function performs
the second phase of the algorithm, and recursively merges the lists generated in the first
phase. In this section we will see how to expose the recursive structure of mergeSort as
an explicit intermediate data structure, and each phase as a recursive morphism with an
associated distributive law. We stress that this structure serves only to turn the recursion
into data, allowing for more explicit control of computation, an idea that is echoed in
our description of two level types, where recursion in a data structure is decomposed.

When considering which data structure with sublinear depth to use for sorting, one
natural choice is the type of balanced binary trees, since these have logarithmic depth in
the number of elements they contain. This tree must faithfully represent the structure of
the sorting algorithm in question; for quicksort, which picks a pivot element and splits
the list in two halves, we would use trees with elements in the branches, where the
element would be the pivot, and each branch a fragment of the list. Mergesort, on the
other hand, works by merging lists at each recursive step. A tree structure with elements
at the leaves is appropriate to encode this behaviour:

data Tree tree = Tip | Leaf Integer | Fork tree tree

instance Functor Tree where
fmap f Tip = Tip

fmap f (Leaf a) = Leaf a
fmap f (Fork l r) = Fork (f l) (f r)

We include a Tip constructor in our datatype that correspond to the empty lists that can
be observed in the output of split. As before, this is a two-level type, and the recursive
site is denoted by the Functor instance.

6.1 First Phase: Growing a Tree

Using the same type directed approach as before, we begin by defining a step function
that relates our two structures. The type of the function we seek is therefore:

grow :: List (Tree× x)→ Tree (List+ x)

Defining such a function for the first few cases is unproblematic, but alas, this line of
development turns out to be in vein, where our best efforts to deal with the following
case are always thwarted:

grow (Cons a (t Leaf b)) = Fork (Go (Cons a ?)) (Stop t)

The goal in this case is to create a Fork that contains the element a in one branch, and b
in the other. The type of Tree demands that the branches in this fork are of type List+ x.
Embedding the b in one branch is a simple matter of making use of Stop t, since t refers
to the Leaf b construct. The problem arises in that we cannot provide a suitable second
parameter to Cons, since the only value of appropriate type x in scope is the t that refers
to the Leaf b constructor. While this is correctly typed, using this value as a parameter
is incorrect for two reasons: first, this would result in a duplication of the value b, and
second, it leads to infinite recursion, since in the next step the value Case a t would
considered again. Possible alternative definitions are either merely variations on this
theme, or satisfy the types by dropping data, which is manifestly unsatisfactory for a
sorting function.

The heart of the problem lies in the fact that the constructors of List offer no way
of signaling the existence of a singleton that should terminate the recursion. In a sense,
there is no constructor in a List that is the counterpart to a Leaf . To recover a step func-
tion that can build trees, the list representation needs a means of expressing singletons
as a primitive constructor. One solution is to lift existing lists so that there is a a new
constructor Single Integer:

data List list = Nil | Single Integer | Cons Integer list

This type gives us everything we need to build the distributive law that relates lists and
trees. The first few cases fall naturally from the types, and there is very little choice in
how to proceed:

grow :: List (Tree× x)→ Tree (List+ x)
grow Nil = Tip
grow (Single a) = Leaf a

grow (Cons a (t Tip)) = Leaf a
grow (Cons a (t Leaf b)) = Fork (Go (Single a)) (Stop t)

Note that here the problem encountered previously has been circumvented by embed-
ding the value a in a Single constructor in the left hand branch of the Fork. In a later
step, this Single a value will be turned into a Leaf a as desired.

The remaining final case where the list contains a Fork can be implemented in a
number of ways. We can arbitrarily insert the element a in either the left or the right
side of the Fork that is produced. Once this choice is established we have another, more
interesting, decision to make with regards to the subtrees given by l and r. One option
is to preserve their order:

grow (Cons a (t Fork l r)) = Fork (Go (Cons a l)) (Stop r)

However, this solution leads to trees that are unbalanced in that new elements are always
inserted on the left side of the tree. The only other option is to reverse the two subtrees
when inserting an element:

grow (Cons a (t Fork l r)) = Fork (Go (Cons a r)) (Stop l)

In so doing, we have rediscovered Braun’s method of producing perfectly balanced trees
(Braun and Rem 1983). In fact, this trick can also be seen in an alternative defintion of
split that considers only empty and non-empty lists, and rotates lists as it recurses.

Since we have described a distributive law, two methods for producing trees emerge:

makeTree,makeTree′ :: Fix List → Fix Tree
makeTree = fold (apo (grow · fmap (id�out)))
makeTree′ = unfold (para (fmap (id� In) ·grow))

The first method, makeTree, encodes the standard way of building a tree by repeated
insertion, using Braun’s method for keeping the tree balanced. The second method,
makeTree′, encodes the slightly more unusual process of generating a tree by repeatedly
uninterleaving a list. This uninterleaving of the list has the same swapping behaviour as
Braun’s method on trees.

6.2 Second Phase: Merging Trees

Once the tree is constructed, the second phase of the algorithm reduces the tree until a
sorted list is produced. The distributive law falls out naturally from the types:

merge :: Tree (List× x)→ List (Tree+ x)
merge Tip = Nil
merge (Leaf a) = Cons a (Go Tip)
merge (Fork (l Nil) (r Nil)) = Nil
merge (Fork (l Nil) (r Cons b r′)) = Cons b (Stop r′)
merge (Fork (l Cons a l′) (r Nil)) = Cons a (Stop l′)
merge (Fork (l Cons a l′) (r Cons b r′))

| a � b = Cons a (Go (Fork l′ r))
| otherwise = Cons b (Go (Fork l r′))

Empty trees give rise to empty lists, and a single element tree produces a single element
list. When we have two subtrees, we inspect the list contained in each subtree. If both
lists are empty, we return the empty list. In case only one of the lists is non-empty, the
element is added to the front of the output sorted list, and the recursion stops with the
tail in hand. In the most general case, we have one element in each branch. We compare
the two elements, picking the smallest one and recursing using the appropriate subtrees.

Note that for this phase, lists do not need to be extended with a Single constructor,
since in the case of Leaf a the fact that the Cons a should terminate is signalled by
embedding a Tip in the tail. This Tip is later turned into a Nil as desired.

As before, we obtain two methods for merging a tree into a list:

mergeTree,mergeTree′ :: Fix Tree → Fix List
mergeTree = fold (apo (merge · fmap (id�out)))
mergeTree′ = unfold (para (fmap (id� In) ·merge))

An operational understanding of how these algorithms work helps develop our under-
standing of the sense in which duality expresses itself here.

The first of these two variations uses a fold in the outer recursion, and this drives
the deconstruction of the tree of values from the bottom until a single list remains. The
controlling fold has an algebra of type:

Tree (Fix List)→ Fix List

Here, the algebra is in fact an apomorphism that starts working at the bottom of the tree.
When a Tip is encountered, the merge will simply produce a Nil, and control is passed
back to the fold. Otherwise, each Leaf a is initially turned into a Cons a (Go Tip), and
the apomorphism continues to unfold the remaining Tip resulting in a remaining struc-
ture that is Cons a Nil. When the fold encounters a Fork, the contents of each branch
are analysed by the apomorphism. In the case where both lists are empty, they become
a single Nil. When only one contains data, that branch is turned into a Cons with this
payload, and the apomorphism is signalled to continue using the appropriate tail with
a Stop. The more interesting case is when both branches contain data. In this case, the
merge function is used to collapse the fork so that the least value is placed at the begin-
ning of the new List. The apomorphism is then signalled with Go to continue merging
the remaining tail of the list that contained the least element, and the whole of the other
list that contained the greater element. This control is directed by constructing a new
Fork that contains these two lists that must be merged, and the Go constructor signi-
fies that the apomorphism must continue merging. Once the apomorphism has finished
merging the lists, control is given back to the fold, which will continue bottom-up, col-
lapsing the tree using the apomorphism, until a single list remains.

The second variation makes use of an unfold that focuses on producing the ensuing
sorted list. The unfold has a coalgebra that has the type:

Fix Tree → List (Fix Tree)

To determine the next element of the ordered list that is to be produced, the unfold must
apply a paramorphism to its seed tree. The work of this paramorphism can be thought
of as collapsing the tree into either a Nil when the tree is empty and the work is finished,
or a Cons a t, where a is the least value in the tree, and t is the tree that is to be used as
the next seed. The tree is collapsed from the bottom, where the paramorphism is applied
recursively at each In Fork until either an In Tip or In (Leaf a) is reached. In Tip values
are turned into Nil values, and In (Leaf a) values are turned into Cons a (In Tip) values.
These intermediate results are then combined bottom-up at each Fork by the function
merge, which maintains the invariant that Cons a t contains the least element a, and t is
the remaining tree. When two such Cons constructors meet at a fork, the least value is
kept, and the seed tree is built out of a new Fork and the appropriate subtrees.

While these two algorithms certainly share many characteristics, their operation
differs significantly. The behaviour of mergeTree is closer to the traditional merge-
sort, where lists are successively merged together until only one remains. In contrast,
mergeTree′ behaves much more like a weak kind of heap sort, where the least ele-
ment is floated out of the tree structure, but no heap property is maintained on the
remaining tree.

6.3 Merging After Growing

Combining these two phases, we can write four variations of mergesort, where simple
functional composition combines the various functions we have already discussed:

mergeSort,mergeSort′,mergeSort′′,mergeSort′′′ :: Fix List → Fix List
mergeSort = mergeTree ·makeTree
mergeSort′ = mergeTree ·makeTree′
mergeSort′′ = mergeTree′ ·makeTree
mergeSort′′′ = mergeTree′ ·makeTree′

In this section we have used the Tree structure as a concrete representation of the
implicit way mergesort works. Generally, however, these intermediate representations
are inefficient, and can be fused away in a process called deforestation (Wadler 1988). A
deforested version of the above algorithms would look similar to mergeSort, as shown
in the beginning of this section. In the fused version, the Tree data structure disappears,
instead becoming implicit from the recursive structure of the function.

7 Conclusion

In this paper we have revisited our previous work on sorting with bialgebras and dis-
tributive laws (Hinze et al. 2012), recasting it in a more applied setting without use of
category theory. Due to the structure of recursive morphisms, and through the use of a
type-directed approach for program construction, we have not lost the intuition behind
our development; the duality is obvious in each sorting method, giving us “algorithms
for free”, and helping to understand the relations between different sorting methods.

Even though we have chosen Haskell as the presentation language for this paper, our
developments readily generalise to other functional programming languages. In partic-
ular, all the code shown compiles with the “exchanging sources” version of the Clean

compiler (Van Groningen et al. 2010) (after some minor refactoring to remove symbolic
operators). This reinforces the argument that sorting algorithms become more clean and
elegant when expressed as distributive laws in recursive morphisms.

